# Κεφάλαιο 4ο: Γραμμική Άλγεβρα

*Όπως ξέρουμε ένας πίνακας εισάγεται είτε με τα στοιχεία του είτε με χρήση της* Table *είτε με χρήση της* Array.

```
a = {{1, 3, 2}, {4, 0, -1}}
b = Table[i^j, {i, 3}, {j, -1, 2}]
c = Array[#1^#2 &, {3, 4}]

{{1, 3, 2}, {4, 0, -1}}
```

$$\left\{\{1, 1, 1, 1\}, \left\{\frac{1}{2}, 1, 2, 4\right\}, \left\{\frac{1}{3}, 1, 3, 9\right\}\right\}$$

```
{{1, 1, 1, 1}, {2, 4, 8, 16}, {3, 9, 27, 81}}
```

Το Array[#1^#2&,{3,4}] *παράγει* 3 *γραμμές με* 4 *στήλες και στοιχεία* $a[i, j] = i^j$. *Για να παράγουμε ακριβώς το* b *θά πρέπει να γράψουμε*

```
Clear[c, d]
c = Array[#1^#2 &, {3, 4}, {1, -1}]
```

$$\left\{\{1, 1, 1, 1\}, \left\{\frac{1}{2}, 1, 2, 4\right\}, \left\{\frac{1}{3}, 1, 3, 9\right\}\right\}$$

Το {1,-1} *στα δεξιά σημαίνει ότι η πρώτη συντεταγμένη ξεκινάει με το* 1 *και η δεύτερη με το* -1. Π.χ

```
Clear[c]
c = Array[d, {3, 4}, {1, -1}]

{{d[1, -1], d[1, 0], d[1, 1], d[1, 2]},
 {d[2, -1], d[2, 0], d[2, 1], d[2, 2]},
 {d[3, -1], d[3, 0], d[3, 1], d[3, 2]}}
```

*Φυσικά, αντί της* Array *μπορούμε να χρησιμοποιήσουμε την* Table *σε κάθε περίπτωση.* Με IdentityMatrix[s] παίρνουμε τον ταυτοτικό πίνακα διαστάσεων s επί s και με DiagonalMatrix[d] παίνουμ εένα διαγώνιο πίνακα με διαγώνια στοιχεία τα στοιχεία της λίστας d. Π.x

```
DiagonalMatrix[{1, 2, 3}]

{{1, 0, 0}, {0, 2, 0}, {0, 0, 3}}
```

# 4.1 Βαθμίδα διανυσμάτων και βαθμίδα (τάξη) πίνακα

*Βαθμίδα των διανυσμάτων* $v_1, v_2, \ldots, v_n$ *ονομάζουμε την διάσταση του γραμμικού χώρου που παράγεται απο τους γραμμικούς συνδυασμούς των διανυσμάτων αυτών. Για να βρούμε την βαθμίδα κάποιων διανυσμάτων εκτελούμε στοιχειώδεις γραμμοπράξεις στον πίνακα με γραμμές τα διανύσματα αυτά(μετάθεση, πρόσθεση ήαφαίρεση κάποιας γραμμής σε μια άλλη κ.ο.κ) έτσι ώστε οι τελευταίες γραμμές να γίνουν μηδενικές και κάθε μη μηδενικήγραμμήνα ξεκινάει με μονάδα- την μονάδα οδηγό. Οι γραμμοπράξεις εκτελούνται με την συνάρτηση* RowReduce. *Το πλήθος των μη μηδενικών γραμμών που προκύπτουν είναι η ζητούμενη βαθμίδα.*

```
RowReduce[a]
RowReduce[b]
```

$\left\{\left\{1, 0, -\frac{1}{4}\right\}, \left\{0, 1, \frac{3}{4}\right\}\right\}$

```
{{1, 0, 0, 6}, {0, 1, 0, -11}, {0, 0, 1, 6}}
```

*Δεν εμφανίζονται μηδενικές γραμμές. Η βαθμίδα τους είναι ίση με* 3. *Άρα και στις δύο περιπτώσεις έχουμε ανεξάρτητα διανύσματα στις γραμμές των* a,b. *Αν προσθέσουμε στην* a *το διάνυσμα* {2,6,4} *τότε χάνεται η γραμ. ανεξαρτησία*:

```
d = Append[a, {2, 6, 4}]
d // MatrixForm
RowReduce[d]

{{1, 3, 2}, {4, 0, -1}, {2, 6, 4}}
```

$\begin{pmatrix} 1 & 3 & 2 \\ 4 & 0 & -1 \\ 2 & 6 & 4 \end{pmatrix}$

$\left\{\left\{1, 0, -\frac{1}{4}\right\}, \left\{0, 1, \frac{3}{4}\right\}, \{0, 0, 0\}\right\}$

*Η μηδενική γραμμή δείχνει την γρμμική εξάρτηση των γραμμών του πίνακα* d (*η τρίτη γραμμήείναι* 2 *φορές την* 1*η*)

Η ***τάξη* r**(d) *ενός πίνακα* d *είναι ίση με το πλήθος των μη μηδενικών γραμμών του* RowReduce[d]. *Στην περίπτωση μας λοιπόν είναι ίση με* 2. *Άρα μόνο δυο απ'αυτές τις γραμμες είναι γρ. ανεξάρτητες.*

Άσκηση: *Δίνεται ένας* 5X5 *πίνακας* a *με γραμμές*

```
x = {1, 2, -1, 0, 1}; y = {2, 1, 0, 1, 3}; z = {0, 3, -2, -1, -1};
t = {2, 4, -2, 0, 2}; s = {4, 5, -2, 1, 5};
```

*Διαπιστώσετε ότι οι γραμμές είναι γρ. εξαρτημένες και στην συνέχεια να βρεθεί ένας μη μηδενικός γραμμικός συνδυασμός τους που να μας δίνει το μηδενικό διάνυσμα.*

***Λύση***: *Θα χρησιμοποιήσουμε την* RowReduce *για να δούμε ότι είναι γρ. εξαρτημένα και στην συνέχεια την* Reduce *ήτην* LinearSolve *για να λύσουμε το σύστημα* a.w=={0,0,0,0,0}. Η Reduce[*εξισώσ,μεταβλ*] *απλοποιεί τις εξισώσ(οι εξισώσ μπορεί να περιλαμβάνουν και ανισώσεις) ως προς τις μεταβλ. Οι εξισώσεις που προκύπτουν είναι ισοδύναμες με τις αρχικές.* Η Reduce[*εξισώσ,μεταβλ,πεδίο*] *περιορίζει την απλοποίηση στο πεδίο(π.χ πεδίο*=Integers)

```
Clear[a]; a = {x, y, z, t, s}; RowReduce[a]
```

$$\left\{\left\{1, 0, \frac{1}{3}, \frac{2}{3}, \frac{5}{3}\right\}, \left\{0, 1, -\frac{2}{3}, -\frac{1}{3}, -\frac{1}{3}\right\}, \{0, 0, 0, 0, 0\}, \{0, 0, 0, 0, 0\}, \{0, 0, 0, 0, 0\}\right\}$$

*Άρα είναι γρ. εξαρτημένες και η τάξη του πίνακα* a *είναι ίση με* 2.

```
Reduce[w0 x + w1 y + w2 z + w3 t + w4 s == 0, {w0, w1, w2, w3, w4}]

w0 == -2 (w2 + w3 + w4) && w1 == w2 - w4
```

Προσέξτε στην Reduce το διάνυσμα στήλη {0,0,0,0,0}στα δεξιά, μπορεί να αντικατασταθεί πολύ απλά και με ένα σκέτο 0 χωρίς να υπάρχει πρόβλημα! *Όμοια με την χρήση της* LinearSolve

```
LinearSolve[a, {0, 0, 0, 0, 0}]

{0, 0, 0, 0, 0}
```

*Παρατηρούμε την διαφορά.* Η LinearSolve *μας έδωσε μόνο μία λύση, την μηδενική*! Το σύνολο των λύσεων ως γνωστό απο τα μαθηματικά (w0,w1,w2,w3,w4) αποτελεί ένα διανυσματικό χώρο που έχει κάποια βάση. Για να βρούμε μια βάση του μπορούμε να χρησιμοποιήσουμε την NullSpace:

```
NullSpace[Transpose[a]]

{{-2, -1, 0, 0, 1}, {-2, 0, 0, 1, 0}, {-2, 1, 1, 0, 0}}
```

Συνεπώς έχει τρία βασικά διανύσματα και άρα διάσταση ίση με 3. Ο παραπάνω γραμμικός χώρος λέγεται μηδενοχώρος(των γραμμών του a) διότι αποτελείται απο εκείνα τα διανύσματα που ικανοποιούν την σχέση Transpose[a].{w0,w1,w2,w3,w4}={0,0,0,0,0}. Παρατηρείστε ότι η διάσταση dim αυτού του χώρου είναι ίση με το πλήθος των **ανεξάρτητων** μεταβλητών της λύσης. Στην περίπτωσή μας οι ανεξάρτητες μεταβλητές είναι 3 (εδώ οι w2, w3, w4) ενώ οι υπόλοιπες 2(w0, w1) γράφονται σαν γραμμικός συνδυασμός των ανεξάρτητων. Γενικά το r[a] μας δίνει το πλήθος των εξαρτημένων μεταβλητών, και η διάσταση του NullSpace[Transpose[a]] το πλήθος των ανεξάρτητων. Τέλος αν n= το πλήθος των αγνώστων(ή αλλιώς το πλήθος των γραμμών του a)τότε μπορούμε άνετα να συμπεράνουμε ότι dim+r[a]=n.

## 4.2 *Γραμμικά συστήματα*

*Έστω ότι έχουμε ένα γραμμικό σύστημα της μορφής* A.X=B *όπου* A *είναι ένας* mXn *πίνακας και* B *είναι ένας* mX1 *πίνακας*. *Το* m *είναι το πλήθος των εξισώσεων και το* n *των αγνώστων*. *Για να έχει λύση θα πρέπει η τάξη του* A *να είναι ίση με την τάξη του επαυξημένου πίνακα* (A|B).
*π.χ για το σύστημα* -2x+y+z=1, x-2y+z=-2, x+y-2z=4 *έχουμε*:

```
A = {{-2, 1, 1},
    {1, -2, 1},
  {1, 1, -2}}; B = {1, -2, 4};
epayxhmenos = {{-2, 1, 1, 1},   {1, -2, 1, -2}, {1, 1, -2, 4}}
RowReduce[A]
RowReduce[epayxhmenos]
```

```
{{-2, 1, 1, 1}, {1, -2, 1, -2}, {1, 1, -2, 4}}
```

```
{{1, 0, -1}, {0, 1, -1}, {0, 0, 0}}
```

```
{{1, 0, -1, 0}, {0, 1, -1, 0}, {0, 0, 0, 1}}
```

*Παρατηρούμε* ότι *οι δύο πίνακες δεν* έχουν *την* ίδια *βαθμίδα*(*τάξη*) άρα *το σύστημα είναι αδύνατον*.

Αυτό *μπρορούμε να τον διαπιστώσουμε και με* άλλο *τρόπο*: *Ζητώντας με την* LinearSolve *να λύσει το σύστημα*:

```
LinearSolve[A, B]
```

```
LinearSolve::nosol : Linear equation encountered which has no solution.
```

```
LinearSolve[{{-2, 1, 1}, {1, -2, 1}, {1, 1, -2}}, {1, -2, 4}]
```

Υπάρχει και *η περίπτωση μιας και **μοναδικής*** *λύσης*. ***Αυτό θα συμβεί*** *όταν ο* A *και ο επαυξημένος* έχουν την ίδια τάξη και *ακριβώς ιση με το πλήθος* των *γραμμών του* A. *Βέβαια για τετραγωνικούς* A *υπάρχει και το κριτήριο της* ορίζουσας: *Άν η* det[A] *είναι μη μηδενική* τότε *κάθε γραμμικό* σύστημα

A.X=B *έχει όπως ξέρουμε μια μοναδική λύση την* $X = A^{-1} B$. Διαφορετικά θα έχει άπειρες λύσεις. *Π.χ ο πίνακας* $\begin{pmatrix} 2 & -1 & 3 \\ 1 & 3 & -2 \\ -1 & 11 & -12 \end{pmatrix}$ *δεν είναι αντιστρέψιμος*:

```
Clear[A]
A = {{2, -1, 3}, {1, 3, -2}, {-1, 11, -12}}; Det[A]

0
```

*οπότε ένα οποιοδήποτε σύστημα με πίνακα συντελεστών τον* A *π.χ* A.X= $\begin{pmatrix} 1 \\ 2 \\ 4 \end{pmatrix}$ θα έχει άπειρες λύσεις:

```
Clear[x, y, z]
Reduce[A.{x, y, z} == {1, 2, 4}, {x, y, z}]
```

$x == \frac{1}{7}(5 - 7z)$ && $y == \frac{1}{7}(3 + 7z)$

```
Clear[x, y, z]; Solve[A.{x, y, z} == {1, 2, 4}, {x, y, z}]

Solve::svars : Equations may not give solutions for all "solve" variables.
```

$\left\{\left\{x \to \frac{5}{7} - z, y \to \frac{3}{7} + z\right\}\right\}$

Η Solve *και η* Reduce *είναι σχετικές*. Η Reduce *γενικά υπερτερεί διότι βρίσκει όλες τις δυνατές λύσεις*.

**Άσκηση**: *Να φτιαχτεί μια συνάρτηση* epayxhmenosMatrix[m_List,k_list] *όπου* m *και* k *είναι δύο πίνακες και που θα επιστρέφει τον επαυξημένο πίνακα τους δηλ. τον πίνακα* m *στον οποίο έχουμε επισυνάψει στα δεξία των στηλών του, τις στήλες του* k. *Υπόδειξη. Χρησιμοποιείστε κατάλληλα την συνάρτηση* Append.

# 4.3 Οι Ιδιοτιμές και τα ιδιοδυανύσματα ενός πίνακα

*Για να βρούμε τα ιδιοδυανύσματα ενός τετραγωνικού πίνακα* A *θα πρέπει πρώτα να βρούμε τις ιδιοτιμές του δηλ. τις ρίζες του χαρατηριστικού πολυωνύμου του* A. *Ξεκινάμε με ένα παράδειγμα.* *Έστω* A= $\begin{pmatrix} 3 & -2 & 0 \\ -2 & 3 & 0 \\ 0 & 0 & 5 \end{pmatrix}$. *Τότε το χαρακτηριστικό πολυώνυμο του είναι ίσο με* Det[A-x IdentityMatrix[3]] *όπου το* IdentityMatrix[3]] *είναι ο ταυτοτικός πίνακας* 3X3

```
Clear[A]
A = ( 3  -2  0
     -2   3  0
      0   0  5 )
charPoly = Det[A - x IdentityMatrix[3]]
idiotimes = Solve[charPoly == 0, x]

{{3, -2, 0}, {-2, 3, 0}, {0, 0, 5}}
```

$25 - 35\,x + 11\,x^2 - x^3$

```
{{x -> 1}, {x -> 5}, {x -> 5}}
```

Με άλλα λόγια *έχουμε δύο ιδιοτιμές την* $\rho_1 = 5$*και την* $\rho_2 = 1$*πολλαπλότητας* 2 *και* 1 *αντίστοιχα*.
*Ένας πιο εύκολος τρόπος να βρίσκουμε τις ιδιοτιμές είναι με την χρήση της* Eigenvalues:

```
Eigenvalues[A]

{1, 5, 5}
```

*Για να βρούμε μια λίστα με τα αντίστοιχα ιδιοδυανύσματα θα χρησιμοποιήσουμε την* Eigenvectors:

```
Eigenvectors[A]

{{1, 1, 0}, {0, 0, 1}, {-1, 1, 0}}
```

Τα δύο τελευταία αντιστοιχούν στην ιδιοτιμή 5 ενώ το πρώτο στην ιδιοτιμή 1. Επίσης για να βρούμε τα ιδιοδιανύσματα μπορούμε να *χρησιμοποιήσουμε την* NullSpace. Η NullSpace[m] όπως έχουμε ήδη αναφέρει *μας δίνει την βάση του χώρου των λύσεων του ομογενούς συστήματος* m.X=**0**. *Οπότε με* NullSpace[A-λIdentityMatrix[n]] (*όπου* n *η διάσταση του* A δηλ. το πλήθος των γραμμών του) *παίρνουμε μια βάση για τα ιδιονύσματα που αντιστοιχούν στην ιδιοτιμή* λ *π.χ.*

```
bashIdioxwroy[5] = NullSpace[A - 5 IdentityMatrix[3]]
bashIdioxwroy[1] = NullSpace[A - 1 IdentityMatrix[3]]

{{0, 0, 1}, {-1, 1, 0}}
```

```
{{1, 1, 0}}
```

*Δηλ. μια βάση του ιδιοχώρου (του χώρου των ιδιοδιανυσμάτων)που αντιστοιχεί στην* λ=5 *είναι η* {{0, 0, 1}, {-1, 1, 0}} *και μια βάση του ιδιοχώρου που αντιστοιχεί στην* λ=1 *είναι η* {{1,1,0}}. Τελειώνουμε *την ενότητα με το αναφέρουμε* ότι *με την συνάρτηση* CharacteristcPolynomial *μπορούμε χωρίς κόπο να βρούμε το χαρακτηριστικό πολυώνυμο*(ως προς κάποια μεταβλητή π.χ. την t):

```
Clear[t]
CharacteristicPolynomial[A, t]
```

$25 - 35\,t + 11\,t^2 - t^3$

## 4.4 Η *διαγωνοποίηση ενός τετραγωνικού πίνακα*

Η *διαγωνοποίηση ενός πίνακα* A *έχει σχέση με τις ιδιοτιμές του και με τους αντίστοιχους ιδιόχωρους*. Είναι *γνωστό απο την θεωρία* ότι *ο* A ***διαγωνοποιείται*** (*δηλ. υπάρχει* ένας *αντιστρέψιμος πίνακας* P *και* ένας *διαγώνιος* D έτσι ώστε D=Inverse[P].A.P) *ανν η πολλαπλότητα της οποιασδήποτε ιδιοτιμής* λ *του* A *συμπίπτει με την διάσταση του αντίστοιχου ιδιοχώρου της*. Άν *κάτι τέτοιο ισχύει τότε* ο A *διαγωνοποιείται και ο* D *έχει στην διαγώνιο τις ιδιοτιμές* του A *και ο* P *έχει στις **στήλες του** τα αντίστοιχα ιδιοδυανύσματα π.χ*

```
P1 = Eigenvectors[A] (*στις γραμμές του P1 τα ιδιοδυανύσματα*)
P = Transpose[P1] (*στις στήλες του P τα ιδιοδυανύσματα*)
Inverse[P]
diagwnios = Inverse[P].A.P // MatrixForm
```

{{1, 1, 0}, {0, 0, 1}, {-1, 1, 0}}

{{1, 0, -1}, {1, 0, 1}, {0, 1, 0}}

$\left\{\left\{\frac{1}{2}, \frac{1}{2}, 0\right\}, \{0, 0, 1\}, \left\{-\frac{1}{2}, \frac{1}{2}, 0\right\}\right\}$

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & 5 \end{pmatrix}$$

Με *την ευκαιρία να αναφέρουμε* ότι *η* DiagonalMatrix[d]*δίνει* ένα *διαγώνιο πίνακα με διαγώνιο* d. *π.χ*

```
DiagonalMatrix[Eigenvalues[A]] // MatrixForm
```

$$\begin{pmatrix} 1 & 0 & 0 \\ 0 & 5 & 0 \\ 0 & 0 & 5 \end{pmatrix}$$

Ο *αντιστρέψιμος πίνακας* P *με την ιδιότητα* P.diagwnios.Inverse[P]=A *δεν υπάρχει πάντα* για κάθε πίνακα A. Αυτό *που* είναι *γνωστό απο την* Γραμμική'Αλγεβρα είναι ότι *υπάρχουν δύο πίνακες* R *και* Q

*και ένας διαγώνιος* D =DiagonalMatrix[m] *έτσι ώστε*:
A=Transpose[R].D.Q. *Οι πίνακες αυτοί μπορούμε να τους βρούμε με την συνάρτηση* SingularValues[A]. H SingularValues[A] ***επιστρέφει** ένα πίνακα* *με στοιχεία* **{R,m,Q}**. *Για να χρησιμοποιήσουμε την* SingularValues[A] *πρέπει τα στοιχεία του* A *να δίνονται με υποδιαστολή* και για αυτό θα πρέπει να χρησιμοποιήσουμε την συνάρτηση N. *Παράδειγμα*:

```
N[A]
b = SingularValues[N[A]]

{{3., -2., 0.}, {-2., 3., 0.}, {0., 0., 5.}}
```

```
{{{-0.707107, 0.707107, 0.},
  {0., 0., 1.}, {-0.707107, -0.707107, 0.}},
 {5., 5., 1.}, {{-0.707107, 0.707107, 0.},
  {0., 0., 1.}, {-0.707107, -0.707107, 0.}}}
```

```
N[A] == Transpose[b[[1]]].DiagonalMatrix[b[[2]]].b[[3]]
b[[1]] // MatrixForm
DiagonalMatrix[b[[2]]] // MatrixForm
b[[3]] // MatrixForm

True
```

$$\begin{pmatrix} -0.707107 & 0.707107 & 0. \\ 0. & 0. & 1. \\ -0.707107 & -0.707107 & 0. \end{pmatrix}$$

$$\begin{pmatrix} 5. & 0 & 0 \\ 0 & 5. & 0 \\ 0 & 0 & 1. \end{pmatrix}$$

$$\begin{pmatrix} -0.707107 & 0.707107 & 0. \\ 0. & 0. & 1. \\ -0.707107 & -0.707107 & 0. \end{pmatrix}$$

# 4.5 Εύρεση δυνάμεων πινάκων

Η διαγωνοποίηση είναι χρήσιμη για την γρήγορη εύρεση δυνάμεων τετραγωνικών πινάκων. Για παράδειγμα στο παραπάνω παράδειγμα υψώνοντας τις ιδιοτιμές στην διαγώνιο εις την 10η μπορούμε να βρούμε την 10η δύναμη του A: $A^{10}$ = P.DiagonalMatrix[{$1^{10}$, $5^{10}$, $5^{10}$}].Inverse[P]

```
P.DiagonalMatrix[{1^10, 5^10, 5^10}].Inverse[P] // MatrixForm
```

$$\begin{pmatrix} 4882813 & -4882812 & 0 \\ -4882812 & 4882813 & 0 \\ 0 & 0 & 9765625 \end{pmatrix}$$

Με την ευκαιρία να αναφέρουμε ότι με A ^ 10παίρνουμε

```
A^10

{{59049, 1024, 0}, {1024, 59049, 0}, {0, 0, 9765625}}
```

δηλ. αποτέλεσμα διαφορετικό απο αυτό που βρήκαμε πριν. Αυτό δεν σημαίνει ότι έχουμε κάνει λάθος παραπάνω. Οφείλεται στο γεγονός ότι στο *Mathematica* ο πολλαπλασιασμός A*A δεν είναι ο γνωστός μας πολλαπλασιασμός πινάκων. Ο γνωστός μας πολλαπλασιαμός πινάκων γίνεται με το Dot[A,B] που συμβολίζεται απλά με A.B Γενικά την n-ιοστή δύναμη του πίνακα A μπορούμε να την ορισουμε αναδρομικά ή αλλιώς μπορούμε να χρησιμοποιήσουμε την συνάρτηση Nest π.χ

```
pollaplasiasmos[a_] := A.a
Nest[pollaplasiasmos, A, 9]

{{4882813, -4882812, 0}, {-4882812, 4882813, 0}, {0, 0, 9765625}}
```

```
?Nest
```

```
Nest[f, expr, n] gives an
  expression with f applied n times to expr.
```

Δηλαδή η Nest επιστρέφει το f(f(...f(expr))...))) όπου το f έχει εφαρμοστεί n φορές στην expr. Στην προηγούμενη χρήση του Nest εφαρμόσαμε 9 φορές το pollaplasiasmos διότι ήδη μέσα στο pollaplasiasmos υπάρχει ήδη 1 εφαρμογή του πολλαπλασιασμού με τον A.

# 4.6 Αλλαγή της βάσης του $\mathbb{R}^n$

*Έστω ότι μας δίνουν δυο βάσεις του* $\mathbb{R}^n$. *Τότε το πέρασμα απο την μία βάση στην άλλη περιγράφεται με ινα αντιστρψιμο πίνακα* P *που λιγεται πίνακας μετάβασης*.Ας δούμε ένα *παράδειγμα*. Δίνεται *μια βάση* $B_1$*του* $\mathbb{R}^4$*και ο πίνακας μετάβασης απο την συνήθη βάση στην* $B_1$*είναι ο* P=$\begin{pmatrix} 1 & 1 & 0 & -1 \\ -1 & 2 & 1 & 0 \\ 2 & -1 & 1 & -2 \\ -2 & -2 & 0 & 3 \end{pmatrix}$.

*Να βρεθούν οι συντεταγμένες του διανύσματος* {1,2,3,4} *στην παλιά(συνήθη) βάση*. *Απάντηση*: *Οι συντεταγμένες είναι το γινόμενο* P.{1,2,3,4}. Άς *δούμε τις πράξεις*

```
P = ( 1   1  0  -1
     -1   2  1   0
      2  -1  1  -2
     -2  -2  0   3 )
```

```
{{1, 1, 0, -1}, {-1, 2, 1, 0}, {2, -1, 1, -2}, {-2, -2, 0, 3}}
```

```
P.{1, 2, 3, 4}
```

```
{-1, 6, -5, 6}
```

*Οι συντεταγμένες του* {1,0,0,0} (*δηλ. του πρώτου βασικού διανύσματος της* $B_1$ ) *στη συνήθη βάση ειναι*:

```
P.{1, 0, 0, 0}
```

```
{1, -1, 2, -2}
```

Αυτή είναι *η πρώτη στήλη του* P. Όμοια *διαπιστώνουμε ότι οι στήλες του* P *είναι οι συντεταγμένες των διανυσμάτων της νέας βάσης ως προς την συνήθη* βάση. Ας δούμε *το αντρίστροφο πρόβλημα* : Δίνεται ένα *διάνυσμα με συντεταγμένες* $\{-1, 6, -5, 6\}$ *ως προς την συνήθη βάση.*Ποιές είναι *οι συντεταγμένες του στην νέα βάση;*

*Σκεφτόμαστε ως εξής επειδή* $\{-1, 6, -5, 6\} = P.\{x_1, x_2, x_3, x_4\}$

*θα πρέπει* $P^{-1}.\{-1, 6, -5, 6\} = \{x_1, x_2, x_3, x_4\}$.

```
Inverse[P].{-1, 6, -5, 6}
```

```
{1, 2, 3, 4}
```

*δηλ. αυτό που περιμέναμε. Γενικά θα πρέπει ο πίνακας μετάβασης* P *από μια βάση σε μια άλλη να είναι ένας αντιστρέψιμος πίνακας. Ας δούμε άλλο ένα παράδειγμα* : *Δίνεται η βάση* $B_2$ *που οι συντεταγμένες των βασικών διανυσμάτων είναι οι στήλες του πίνακα* **Q =**

$$\begin{pmatrix} -2 & 0 & -2 & 1 \\ -1 & 1 & 0 & -2 \\ 1 & 2 & -1 & -1 \\ -2 & 2 & 1 & -2 \end{pmatrix}$$

. *Αφού δείξετε ότι πράγματι αποτελούν βάση να βρείτε τον πίνακα μετάβασης απο την* $B_1$ *στη* $B_2$.

*Απάντηση* : *Κατ' αρχήν θα ελέγξουμε την ορίζουσα του* Q. *Αν είναι μη μηδενική τότε οι στήλες είναι γραμμικώς ανεξάρτητες και άρα αποτελούν μια βάση. Για να βρούμε τον πίνακα μετάβασης απο την* $B_1$ *στη* $B_2$ *βρίσκουμε πρώτα τον πίνακα μετάβασης απο την* $B_1$ *στην συνήθη (αυτός είναι ο* Inverse[P]) *και απο την συνήθη στην* $B_2$ (*μέσω του* Q). *Τελικά ο ζητούμενος είναι ο* Inverse[P].Q

```
Q = (-2  0  -2   1
     -1  1   0  -2
      1  2  -1  -1
     -2  2   1  -2)
```

```
{{-2, 0, -2, 1}, {-1, 1, 0, -2}, {1, 2, -1, -1}, {-2, 2, 1, -2}}
```

```
Det[Q]
```

```
25
```

```
Inverse[P]
```

$$\left\{\left\{\frac{13}{6}, -\frac{1}{6}, \frac{1}{6}, \frac{5}{6}\right\}, \left\{\frac{5}{6}, \frac{1}{6}, -\frac{1}{6}, \frac{1}{6}\right\}, \left\{\frac{1}{2}, \frac{1}{2}, \frac{1}{2}, \frac{1}{2}\right\}, \{2, 0, 0, 1\}\right\}$$

```
MatrixForm[Inverse[P].Q]
```

$$\begin{pmatrix} -\frac{17}{3} & \frac{11}{6} & -\frac{11}{3} & \frac{2}{3} \\ -\frac{7}{3} & \frac{1}{6} & -\frac{4}{3} & \frac{1}{3} \\ -2 & \frac{5}{2} & -1 & -2 \\ -6 & 2 & -3 & 0 \end{pmatrix}$$

*Γενικά μπορούμε να φτιάξουμε μια συνάρτηση με είσοδο δυο δοθείσες βάσεις* $V_1, V_2$(*ήνα το πούμε καλύτερα με είσοδο τις συντεταγμένες των βασικών διανυσμάτων των* $V_1, V_2$ *ώς προς την συνήθη βάση*)*και θα μας επιστρέφει τον πίνακα μετάβασης απο την μια στην άλλη.*

```
changeBasis[V1_List, V2_List] :=
 If[Det[V1] != 0, Inverse[Transpose[V1]].Transpose[V2],
  "η αλλαγήβάσεως δεν είναι δυνατή"]
```

$$\text{changeBasis}\left[\begin{pmatrix} 1 & -1 & 2 & -2 \\ 1 & 2 & -1 & -2 \\ 0 & 1 & 1 & 0 \\ -1 & 0 & -2 & 3 \end{pmatrix}, \begin{pmatrix} -2 & -1 & 1 & -2 \\ 0 & 1 & 2 & 2 \\ -2 & 0 & -1 & 1 \\ 1 & -2 & -1 & -2 \end{pmatrix}\right]$$

$$\left\{\left\{-\frac{17}{3}, \frac{11}{6}, -\frac{11}{3}, \frac{2}{3}\right\}, \left\{-\frac{7}{3}, \frac{1}{6}, -\frac{4}{3}, \frac{1}{3}\right\}, \left\{-2, \frac{5}{2}, -1, -2\right\}, \{-6, 2, -3, 0\}\right\}$$

**Άσκηση:** O *προσαρτημένος πίνακας*(adjoint) *ενός τετραγωνικού πίνακα* A *συμβολίζεται με* adj[A] *και έχει στοιχεία τα* $b[i, j] = (-1)^{i+j} D_{j,i}$*όπου με* $D_{j,i}$*συμβολίζουμε την ορίζουσα του* A *όταν απο τον* A *αφαιρεθεί η* j *γραμμήκαι η* i *στήλη*. *Κατασκευάσετε μια συνάρτηση* minorMatrix[m_List,i_Integer/;-Positive[i],j_Integer/;Positive[j]] *που με είσοδο τα* m, i,j *θα επιστρέφει τον ελλάσονα πίνακα του* m *χωρίς την* i *γραμμήκαι την* j *στήλη*. *Στην συνέχεια κατασκευάσετε την συνάρτηση* adjointMatrix[m] *που θα επιστρέφει τον προσαρτημένο πίνακα του* m *και δοκιμάστε*( *για ένα συγκεκριμένο* m)*αν ισχύει η ισότητα*:
m.adjointMatrix[m] ==Det[m].IdentityMatrix[Length[m]]==adjointMatrix[m].m *Υπόδειξη*: *Για την κατασκευήτης* minorMatrix *μπορείτε να χρησιμοποιήσετε τις συναρτήσεις* Drop *και* Part *που είδαμε σε προηγούμενο μάθημα*.

# 4.7 Γραμμικές συναρτήσεις και πίνακες

*Σε αυτήν την ενότητα θα μελετήσουμε τους πίνακες απο μια άλλη σκοπιά*. *Κάθε πίνακας* A *διαστάσεων* mXn *ορίζει μια γραμμικήσυνάρτηση* $f : \mathbb{R}^n -> \mathbb{R}^m$*με ορισμό* $f[e]$=A.$\{x_1, .. x_n\}$. (*με* $\{x_1, .. x_n\}$ *εννοούμε* ***τις συντεταγμένες*** *του διανύσματος* e *του* $\mathbb{R}^n$ *ως προς την συνήθη βάση του* $\mathbb{R}^n$. *Ακόμα με* f[e] *δεν εννοούμε κάποιο διάνυσμα* $e_1$*του* $\mathbb{R}^m$*αλλά* ***τις συντεταγμένες*** *του* $e_1$ *ως προς την συνήθη βάση του* $\mathbb{R}^m$ ). *Αντίστροφα κάθε γραμμική*$f : \mathbb{R}^n -> \mathbb{R}^m$ *αντιστοιχεί σε ένα πίνακα* A. *Ας δούμε ένα παράδειγμα*:

$$\mathbf{A} = \begin{pmatrix} 2 & 3 & -1 \\ 3 & -1 & 2 \\ 1 & 2 & 3 \end{pmatrix}$$

```
{{2, 3, -1}, {3, -1, 2}, {1, 2, 3}}
```

*τότε η συνάρτηση* matrixToFunction *που ορίζεται παρακάτω τον μετατρέπει σε γραμμικήσυνάρτηση*:

```
matrixToFunction[m_List] := m.Table[x_i, {i, Length[First[m]]}]
f = matrixToFunction[A]
```

$$\{2\, x_1 + 3\, x_2 - x_3,\ 3\, x_1 - x_2 + 2\, x_3,\ x_1 + 2\, x_2 + 3\, x_3\}$$

Το Length[First[m]] *είναι ίσο με το πλήθος των στηλών του* m. *Για το αντίστροφο πρόβλημα πρέπει να φτιάξουμε μια συνάρτηση* functionToMatrix *που να μας επιστρέφει τον πίνακα που κρύβεται πίσω απο μια γραμμικήσυνάρτηση* f. *Για τον σκοπό αυτό θα χρειαστούμε την* Variables[f] *που επιστρέφει τις μεταβλητές της* f *και την* Coefficient[g,*λίστα*] *που δίνει τους συντελεστές των μεταβλητών*(ήτων *δυνάμεων μεταβλητών*) *της λίστα στην συνάρτηση* g. Π.*χ*

```
Coefficient[2 x + 5 y^2, {x}]
Coefficient[2 x + 5 y^2, {x, y}]
Coefficient[2 x + 5 y y, {x, y^2}]

{2}
```

```
{2, 0}
```

```
{2, 5}
```

Παρατηρείστε ότι *στο πολυώνυμο* 2 x +5 $y^2$ *ο* Coefficient *του* y *είναι* 0(*και* όχι 5 y) *ενώ του* $y^2$*είναι* 5! Η functionToMatrix *ορίζεται ως εξής*:

```
functionToMatrix[f_List] :=
 Table[Coefficient[f[[i]], Variables[f]], {i, Length[f]}]
functionToMatrix[f]

{{2, 3, -1}, {3, -1, 2}, {1, 2, 3}}
```

**Άσκηση**: *Δίνεται η γραμμικήσυνάρτηση* f(e)={x+2y, y-x,2x} *όπου* x, y, z *είναι οι συντεγμένες του* e *ως προς την συνήθη βάση*. *Βρείτε τον τύπο της* f *όταν αλλάξουμε την συνήθη βάση του* $\mathbb{R}^2$*στην βάση* $B = \{v_1, v_2\}$ *όπου* $v_1$={1,1} *και* $v_2 = \{1, 2\}$ *και την βάση του* $\mathbb{R}^3$(*του πεδίου τιμών της* f ) *στην* $B^* = \{u_1, u_2, u_3\}$*όπου* $u_1 = \{1, 1, 1\}$, $u_2 = \{1, 1, 0\}$, $u_3 = \{1, 0, 0\}$.

*Υπόδειξη*: *Έστω ενα τυχαίο διάνυσμα* q *γραμμένο ως προς την βάση* B. *Οι συντεταγμένες του ως προς την συνήθη βάση είναι* $e^* = \{x^*, y^*, z^*\}$=Transpose[B].q *Όπότε με* $f(e^*)$*βρίσκομε τις συντεταγμένες της εικόνος*(*ως προς την συνήθη βάση*) *και με* Inverse[Transpose[$B^*$]].$f(e^*)$ *βρίσκουμε τις ζητούμενες συντεταγμένες ως προς την βάση* $B^*$.

*Άλλος τρόπος*: *έστω* A *ο πίνακας της* f *ως προς τις συνήθεις βάσεις και έστω* P = transpose[B]*ο πίνακας μετάβασης απο την συνήθη βάση του* $\mathbb{R}^2$*στην* B *και έστω* $Q$ = Transpose[$B^*$]*ο πίνακας μετάβασης απο την συνήθη βάση του* $\mathbb{R}^3$*στην* $B^*$. *Τότε είναι γνωστό απο την θεωρία ότι ο πίνακας της* f *ως προς τις βάσεις* B *στην* $B^*$*είναι ίσος με* $A^* = Q^{-1}.A.P$. *Απο εδώ μπορούμε εύκολα να βρούμε τον νέο τύπο της* f *με ένα απλό πολλαπλασιασμό* $f^*(q) = A^*.q$